```
> restart:
  with(LinearAlgebra):
  d:=1;
```

$$d := 1 \tag{1}$$

We define the Lax matrix using Theorem 2.2

```
> checkL12:=-t[2*d+1]*lambda^(2*d):
  for k from 0 to 2*d-1 do checkL12:=checkL12-t[2*d+1]*Qinfty[k]*
  lambda^k: od:
  checkL12:=checkL12;
```

$$checkL12 := -t_3\,\lambda^2 - t_3\,Qinfty_1\,\lambda - t_3\,Qinfty_0 \tag{2}$$

```
> checkL11:=t00/lambda+t00/Qinfty[0]*lambda^(2*d-1):
  for k from 0 to 2*d-1 do checkL11:=checkL11-Pinfty[2*d-1-k]*
  lambda^k: od:
  for k from 0 to 2*d-2 do for m from 0 to 2*d-2-k do checkL11:=
  checkL11-Pinfty[m]*Qinfty[k+1+m]*lambda^k: od: od:
  for k from 1 to 2*d-1 do checkL11:=checkL11+t00/Qinfty[0]*Qinfty
  [k]*lambda^(k-1): od:
  checkL11:=checkL11;

  checkL22:=-checkL11:
```

$$checkL11 := \frac{t00}{\lambda} + \frac{t00\,\lambda}{Qinfty_0} - Pinfty_1 - Pinfty_0\,\lambda - Pinfty_0\,Qinfty_1 + \frac{t00\,Qinfty_1}{Qinfty_0} \tag{3}$$

```
> NumcheckL21poleinfinity:=-checkL11^2/checkL12:
  for j from 2*d-1 to 4*d do for m from 0 to 4*d-j do
  NumcheckL21poleinfinity:=NumcheckL21poleinfinity+t[2*d+1-m]*t[j+
  m-2*d+1]*lambda^j/checkL12: od: od:
  checkL21poleinfinity:=0:
  for i from 1 to 100 do
  checkL21poleinfinity:=checkL21poleinfinity-(residue
  (NumcheckL21poleinfinity*lambda^(-i),lambda=infinity))*lambda^
  (i-1): od:

  checkL21polezero:=simplify(-(residue(NumcheckL21poleinfinity,
  lambda=infinity))/lambda):
  checkL21:=checkL21polezero+checkL21poleinfinity;
```

$$checkL21 := \frac{1}{Qinfty_0^2\,t_3\,\lambda}\Big(\big(-2\,t_3^2\,Qinfty_1 + 2\,t_3\,t_2\big)\,Qinfty_0^3 + \big(Qinfty_1^3\,t_3^2 - 2\,t_2\,Qinfty_1^2\,t_3 + \big( \tag{4}$$
$$Pinfty_0^2 + 2\,t_3\,t_1 + t_2^2\big)\,Qinfty_1 - 2\,t_3\,t_0 - 2\,t_2\,t_1 + 2\,Pinfty_0\,Pinfty_1\big)\,Qinfty_0^2$$

$$-2\,t00\left(Pinfty_0\,Qinfty_1+Pinfty_1\right)Qinfty_0+t00^2\,Qinfty_1\Big)+\frac{\left(\frac{t00}{Qinfty_0}-Pinfty_0\right)^2}{t_3}-2\,t_1$$
$$-\frac{t_2^2}{t_3}+2\,t_2\,Qinfty_1-\frac{t_3^2\,Qinfty_1^2-t_3^2\,Qinfty_0}{t_3}-\left(-t_3\,Qinfty_1+2\,t_2\right)\lambda-t_3\,\lambda^2$$

```
> checkLMatrix:=Matrix(2,2,0):
  checkLMatrix[1,1]:=checkL11:
  checkLMatrix[1,2]:=checkL12:
  checkLMatrix[2,1]:=checkL21:
  checkLMatrix[2,2]:=checkL22:
  checkLMatrix:
```

We define the general auxiliary matrix using Theorem 2.2

```
> checkA12:=0:
  for i from 0 to 2*d-1 do checkA12:=checkA12-t[2*d+1]*nu[i]*
  lambda^(2*d-i): od:
  for k from 1 to 2*d-1 do for i from 0 to k-1 do checkA12:=
  checkA12-t[2*d+1]*nu[i]*Qinfty[k]*lambda^(k-i): od: od:
  checkA12;

  nu[2*d]:=simplify(-residue(lambda^(2*d-1)*checkA12/checkL12,
  lambda=infinity));
```

$$-t_3\,\nu_0\,\lambda^2-t_3\,\nu_0\,Qinfty_1\,\lambda-t_3\,\nu_1\,\lambda$$
$$\nu_2:=-Qinfty_0\,\nu_0-Qinfty_1\,\nu_1 \tag{5}$$

```
> checkA11:=0:
  NumcheckA11:=0:
  for i from 0 to 2*d-1 do NumcheckA11:=NumcheckA11+ nu[i]*lambda^
  (-i)*checkL11: od:
  for i from 1 to 100 do  checkA11:=checkA11-(residue(NumcheckA11*
  lambda^(-i),lambda=infinity))*lambda^(i-1): od:
  checkA11;

  checkA22:=-checkA11:
```

$$\nu_0\left(-Pinfty_1-Pinfty_0\,Qinfty_1+\frac{t00\,Qinfty_1}{Qinfty_0}\right)+\nu_1\left(\frac{t00}{Qinfty_0}-Pinfty_0\right)+\nu_0\left(\frac{t00}{Qinfty_0}-Pinfty_0\right)\lambda \tag{6}$$

```
> checkA21:=(-t00)/t[2*d+1]/(Qinfty[0])^2*LQinfty[0]/lambda:
  QuantitycheckA21:=checkL11/checkL12*(checkL11/checkL12*checkA12
  -2*checkA11):
```

```
for i from 1 to 100 do  checkA21:=checkA21-(residue
(QuantitycheckA21*lambda^(-i),lambda=infinity))*lambda^(i-1): od:
for i from 0 to 100 do  checkA21:=checkA21+(residue
(QuantitycheckA21*lambda^(i),lambda=0))*lambda^(-i-1): od:
AdditionalTerm:=t00*(1-t00)/t[2*d+1]/Qinfty[0]^2/lambda*nu[2*d-1]
:
for k from 1 to 2*d-1 do AdditionalTerm:=AdditionalTerm+t00*(1-
t00)/t[2*d+1]/Qinfty[0]^2/lambda*nu[k-1]*Qinfty[k]: od:
checkA21:=checkA21+AdditionalTerm:
NumcheckA21:=0:
for k from 2*d to 4*d do for j from k to 4*d do for m from 0 to
4*d-j do NumcheckA21:=NumcheckA21- t[2*d+1-m]*t[j+m-2*d+1]*nu[j-
k]*lambda^k: od: od: od:
for i from 1 to 100 do  checkA21:=checkA21-(residue(NumcheckA21/
(-checkL12)*lambda^(-i),lambda=infinity))*lambda^(i-1): od:
checkA21:=checkA21:
series(checkA21,lambda=infinity);
```

$$-t_3 \nu_0 \lambda^2 + \frac{\left(t_3^2 \nu_0 \mathit{Qinfty}_1 - 2\, t_3 t_2 \nu_0 - t_3^2 \nu_1\right) \lambda}{t_3} + \frac{\left(\frac{\mathit{t00}}{\mathit{Qinfty}_0} - \mathit{Pinfty}_0\right)^2 \nu_0}{t_3} + \frac{1}{t_3}\Big(-2\, t_3 t_1 \nu_0 - t_2^2 \nu_0 - 2\, t_3 t_2 \nu_1 - t_3^2 \left(-\mathit{Qinfty}_0 \nu_0 - \mathit{Qinfty}_1 \nu_1\right) + t_3^2 \nu_0 \mathit{Qinfty}_0 - \left(t_3 \nu_0 \mathit{Qinfty}_1 - 2\, \nu_0 t_2 - t_3 \nu_1\right) t_3 \mathit{Qinfty}_1\Big) + \frac{1}{\lambda}\Bigg(-\frac{\mathit{t00}\, \mathit{LQinfty}_0}{t_3 \mathit{Qinfty}_0^2} + \frac{1}{\mathit{Qinfty}_0^4 t_3}\Big(-2\, \mathit{t00}\, \mathit{Pinfty}_0 \mathit{Qinfty}_0^3 \mathit{Qinfty}_1 \nu_0 + \mathit{t00}^2 \mathit{Qinfty}_0^2 \mathit{Qinfty}_1 \nu_0 - 2\, \mathit{t00}\, \mathit{Pinfty}_0 \mathit{Qinfty}_0^3 \nu_1 - 2\, \mathit{t00}\, \mathit{Pinfty}_1 \mathit{Qinfty}_0^3 \nu_0 + \mathit{t00}^2 \mathit{Qinfty}_0^2 \nu_1\Big) + \frac{\mathit{t00}\,(1-\mathit{t00})\, \nu_1}{t_3 \mathit{Qinfty}_0^2} + \frac{\mathit{t00}\,(1-\mathit{t00})\, \nu_0 \mathit{Qinfty}_1}{t_3 \mathit{Qinfty}_0^2}\Bigg) \quad \textbf{(7)}$$

```
> checkAMatrix:=Matrix(2,2,0):
  checkAMatrix[1,1]:=checkA11:
  checkAMatrix[1,2]:=checkA12:
  checkAMatrix[2,1]:=checkA21:
  checkAMatrix[2,2]:=checkA22:
  checkAMatrix:
```

We define the derivatives of the Lax matrices. The derivative relatively to times and Darboux coordinates is specific to d=1 and should be adapted for other values of d.

```
> dcheckAMatrixdlambda:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dcheckAMatrixdlambda[i,
  j]:=diff(checkAMatrix[i,j],lambda): od: od:
> LcheckLMatrix:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do
```

```
LcheckLMatrix[i,j]:=diff(checkLMatrix[i,j],t[2])*alpha[2]+diff
(checkLMatrix[i,j],t[1])*alpha[1]+diff(checkLMatrix[i,j],Qinfty
[0])*LQinfty[0]+diff(checkLMatrix[i,j],Qinfty[1])*LQinfty[1]+diff
(checkLMatrix[i,j],Pinfty[0])*LPinfty[0]+ diff(checkLMatrix[i,j],
Pinfty[1])*LPinfty[1]: od:  od:
```

We define the compatibility equations and solve them. This is the main test for the formulas of Theorem 2.2. This part is again specific to d=1 and should be adapted for other values of d

```
> CompatibilityEquations:=simplify(
dcheckAMatrixdlambda-LcheckLMatrix+Multiply
(checkAMatrix,checkLMatrix)-Multiply(checkLMatrix,checkAMatrix)
):
```

```
> simplify(series(CompatibilityEquations[1,1],lambda=0)):
series(simplify(series(CompatibilityEquations[1,1],lambda=
infinity)),lambda=infinity):
series(simplify(series(CompatibilityEquations[1,2],lambda=
infinity)),lambda=infinity):
series(simplify(series(CompatibilityEquations[2,1],lambda=
infinity)),lambda=infinity):
EQ1:=residue(CompatibilityEquations[1,1],lambda=0);
EQ2:=-residue(CompatibilityEquations[1,1]/lambda^2,lambda=
infinity);
EQ3:=-residue(CompatibilityEquations[1,1]/lambda^1,lambda=
infinity);
EQ4:=-residue(CompatibilityEquations[1,2]/lambda^2,lambda=
infinity);
EQ5:=-residue(CompatibilityEquations[1,2]/lambda^1,lambda=
infinity);
```

$$EQ1 := \frac{1}{Qinfty_0^2}\left(-2\, t00\, Pinfty_0\, Qinfty_0^2\, Qinfty_1\, \nu_0 - 2\, t00\, Pinfty_0\, Qinfty_0^2\, \nu_1 - 2\, t00\, Pinfty_1\, Qinfty_0^2\, \nu_0 + t00\, Qinfty_0\, Qinfty_1\, \nu_0 - t00\, LQinfty_0\, Qinfty_0 + t00\, Qinfty_0\, \nu_1\right)$$

$$EQ2 := \frac{1}{Qinfty_0^2}\Big(\big(\big(4\, Qinfty_1\, t_3^2 - 4\, t_2\, t_3\big)\, \nu_0 - 2\, t_3^2\, \nu_1\big)\, Qinfty_0^3 + \Big(\big(-Qinfty_1^3\, t_3^2 + 2\, t_2\, Qinfty_1^2\, t_3 + \big(-Pinfty_0^2 - 2\, t_1\, t_3 - t_2^2\big)\, Qinfty_1 - 2\, Pinfty_0\, Pinfty_1 + 2\, t_3\, t_0 + 2\, t_2\, t_1\big)\, \nu_0 + 3\, Qinfty_1^2\, t_3^2\, \nu_1 - 4\, t_2\, t_3\, \nu_1\, Qinfty_1 + \big(-Pinfty_0^2 + 2\, t_1\, t_3 + t_2^2\big)\, \nu_1 + LPinfty_0\Big)\, Qinfty_0^2 - t00\, \big(Qinfty_1\, \nu_0 + \nu_1\big)\, (-1 + t00)\Big)$$

$$EQ3 := \frac{1}{Qinfty_0^2}\Big(2\, \nu_0\, t_3^2\, Qinfty_0^4 + \Big(\big(Qinfty_1^2\, t_3^2 + Pinfty_0^2 - 2\, t_1\, t_3 - t_2^2\big)\, \nu_0 + 4\, t_3\, \nu_1\, \big(t_3\, Qinfty_1 - t_2\big)\Big)\, Qinfty_0^3 + \Big(\big(-Qinfty_1^4\, t_3^2 + 2\, Qinfty_1^3\, t_2\, t_3 + \big(-Pinfty_0^2 - 2\, t_1\, t_3 - t_2^2\big)\, Qinfty_1^2 + \big(-2\, Pinfty_0\, Pinfty_1 + 2\, t_3\, t_0 + 2\, t_2\, t_1\big)\, Qinfty_1 - 2\, t00\, Pinfty_0 - Pinfty_0\big)\, \nu_0 - Qinfty_1^3\, t_3^2\, \nu_1 + 2$$

$$Qinfty_1^2\, t_2\, t_3\, \nu_1 + \left( \left( -Pinfty_0^2 - 2\, t_1\, t_3 - t_2^2 \right) \nu_1 + LPinfty_0 \right) Qinfty_1 + \left( -2\, Pinfty_0\, Pinfty_1 + 2\, t_3\, t_0 + 2\, t_2\, t_1 \right) \nu_1 + LQinfty_1\, Pinfty_0 + LPinfty_1 \right) Qinfty_0^2 + \left( (t00 + 1)\, \nu_0 + 2\, Pinfty_1\, \nu_1 - LQinfty_1 \right) t00\, Qinfty_0 - t00\, Qinfty_1 \left( Qinfty_1\, \nu_0 + \nu_1 \right) (-1 + t00) \right)$$

$$EQ4 := t_3 \left( \left( 2\, Pinfty_0\, Qinfty_0 - 2 \right) \nu_0 - 2\, Pinfty_1\, \nu_1 + LQinfty_1 \right)$$

$$EQ5 := t_3 \left( \left( \left( 2\, Pinfty_0\, Qinfty_1 + 2\, Pinfty_1 \right) Qinfty_0 - Qinfty_1 \right) \nu_0 + 2\, Pinfty_0\, Qinfty_0\, \nu_1 - \nu_1 + LQinfty_0 \right) \tag{8}$$

```
> LQinfty[0]:=factor(solve(EQ1,LQinfty[0]));
  LQinfty[1]:=factor(solve(EQ4,LQinfty[1]));
  LPinfty[0]:=factor(solve(EQ2,LPinfty[0]));
  LPinfty[1]:=factor(solve(EQ3,LPinfty[1]));
  EQ1:=simplify(EQ1);
  EQ2:=simplify(EQ2);
  EQ3:=simplify(EQ3);
  EQ4:=simplify(EQ4);
  EQ5:=simplify(EQ5);
```

$$LQinfty_0 := -2\, Pinfty_0\, Qinfty_0\, Qinfty_1\, \nu_0 - 2\, Pinfty_0\, Qinfty_0\, \nu_1 - 2\, Pinfty_1\, Qinfty_0\, \nu_0 + Qinfty_1\, \nu_0 + \nu_1$$

$$LQinfty_1 := -2\, Pinfty_0\, Qinfty_0\, \nu_0 + 2\, Pinfty_1\, \nu_1 + 2\, \nu_0$$

$$LPinfty_0 := \frac{1}{Qinfty_0^2} \left( Qinfty_0^2\, Qinfty_1^3\, \nu_0\, t_3^2 - 4\, Qinfty_0^3\, Qinfty_1\, \nu_0\, t_3^2 - 2\, Qinfty_0^2\, Qinfty_1^2\, \nu_0\, t_2\, t_3 - 3\, Qinfty_0^2\, Qinfty_1^2\, \nu_1\, t_3^2 + Pinfty_0^2\, Qinfty_0^2\, Qinfty_1\, \nu_0 + 4\, Qinfty_0^3\, \nu_0\, t_2\, t_3 + 2\, Qinfty_0^3\, \nu_1\, t_3^2 + 2\, Qinfty_0^2\, Qinfty_1\, \nu_0\, t_1\, t_3 + Qinfty_0^2\, Qinfty_1\, \nu_0\, t_2^2 + 4\, Qinfty_0^2\, Qinfty_1\, \nu_1\, t_2\, t_3 + Pinfty_0^2\, Qinfty_0^2\, \nu_1 + 2\, Pinfty_0\, Pinfty_1\, Qinfty_0^2\, \nu_0 - 2\, Qinfty_0^2\, \nu_0\, t_0\, t_3 - 2\, Qinfty_0^2\, \nu_0\, t_1\, t_2 - 2\, Qinfty_0^2\, \nu_1\, t_1\, t_3 - Qinfty_0^2\, \nu_1\, t_2^2 + t00^2\, Qinfty_1\, \nu_0 + t00^2\, \nu_1 - t00\, Qinfty_1\, \nu_0 - t00\, \nu_1 \right)$$

$$LPinfty_1 := -\frac{1}{Qinfty_0} \left( -3\, Qinfty_0^2\, Qinfty_1^2\, \nu_0\, t_3^2 - 4\, Qinfty_0\, Qinfty_1^3\, \nu_1\, t_3^2 + 2\, t_3^2\, Qinfty_0^3\, \nu_0 + 4\, Qinfty_0^2\, Qinfty_1\, \nu_0\, t_2\, t_3 + 6\, Qinfty_0^2\, Qinfty_1\, \nu_1\, t_3^2 + 6\, Qinfty_0\, Qinfty_1^2\, \nu_1\, t_2\, t_3 - Pinfty_0^2\, Qinfty_0^2\, \nu_0 - 2\, Qinfty_0^2\, \nu_0\, t_1\, t_3 - Qinfty_0^2\, \nu_0\, t_2^2 - 4\, Qinfty_0^2\, \nu_1\, t_2\, t_3 - 4\, Qinfty_0\, Qinfty_1\, \nu_1\, t_1\, t_3 - 2\, Qinfty_0\, Qinfty_1\, \nu_1\, t_2^2 + 2\, Qinfty_0\, \nu_1\, t_0\, t_3 + 2\, Qinfty_0\, \nu_1\, t_1\, t_2 + t00^2\, \nu_0 + Pinfty_0\, Qinfty_0\, \nu_0 - t00\, \nu_0 \right)$$

$$EQ1 := 0$$
$$EQ2 := 0$$
$$EQ3 := 0$$
$$EQ4 := 0$$
$$EQ5 := 0 \tag{9}$$

```
> series(simplify(series(CompatibilityEquations[2,1],lambda=
  infinity)),lambda=infinity);
```

(10)

$$\frac{\left(-4\, t_3^2\, \nu_0 + 2\, \alpha_2\, t_3\right) \lambda}{t_3} + \frac{\left(4\, Qinfty_1\, \nu_0 - 2\, \nu_1\right) t_3^2 + \left(-2\, Qinfty_1\, \alpha_2 - 6\, \nu_0\, t_2 + 2\, \alpha_1\right) t_3 + 2\, \alpha_2\, t_2}{t_3} + \frac{1}{t_3\, \lambda}\Big(\Big(-4\, Qinfty_1^2\, \nu_0 + 4\, Qinfty_0\, \nu_0 + 2\, Qinfty_1\, \nu_1\Big) t_3^2 + \Big(2\, Qinfty_1^2\, \alpha_2 + \left(6\, \nu_0\, t_2 - 2\, \alpha_1\right) Qinfty_1 - 4\, \nu_0\, t_1 - 2\, \nu_1\, t_2 - 2\, Qinfty_0\, \alpha_2\Big) t_3 - 2\, Qinfty_1\, \alpha_2\, t_2 - 2\, t_2^2\, \nu_0 + 2\, \alpha_2\, t_1 + 2\, \alpha_1\, t_2\Big) \quad \textbf{(10)}$$

In order to solve the entry (2,1) of the compatibility equations, we need to express the nu's in terms of the vector \alpha.

```
> solve(-4*nu[0]*t[3]^2+2*alpha[2]*t[3],nu[0]);
  solve(2*nu[1]*t[3]^2-2*alpha[1]*t[3]+alpha[2]*t[2],nu[1]);
```

$$\frac{\alpha_2}{2\, t_3}$$

$$\frac{2\, \alpha_1\, t_3 - \alpha_2\, t_2}{2\, t_3^2} \quad \textbf{(11)}$$

```
> M:=Matrix(2*d,2*d,0):
  for i from 1 to 2*d do for j from 1 to i do
  M[i,j]:=t[2*d+1-i+j]:
  od: od:
  M;
  alphaVector:=Matrix(2*d,1,0):
  for i from 1 to 2*d do alphaVector[i,1]:=alpha[2*d-i+1]/(2*d-i+1)
  : od:
  alphaVector;
  nuVector:=Matrix(2*d,1,0):
  for i from 1 to 2*d do nuVector[i,1]:=nu[i-1]: od:
  nuVector;
```

$$\begin{bmatrix} t_3 & 0 \\ t_2 & t_3 \end{bmatrix}$$

$$\begin{bmatrix} \frac{\alpha_2}{2} \\ \alpha_1 \end{bmatrix}$$

$$\begin{bmatrix} \nu_0 \\ \nu_1 \end{bmatrix} \quad \textbf{(12)}$$

```
> nu[0]:=alpha[2]/(2*t[3]);
  nu[1]:=(2*alpha[1]*t[3]-alpha[2]*t[2])/(2*t[3]^2);
```

```
simplify(Multiply(M,nuVector)-alphaVector);
```

$$\nu_0 := \frac{\alpha_2}{2\,t_3}$$

$$\nu_1 := \frac{2\,\alpha_1\,t_3 - \alpha_2\,t_2}{2\,t_3^2}$$

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \qquad (13)$$

```
> simplify(CompatibilityEquations);
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \qquad (14)$$

This verifies that the relation between nu's and alpha's is correct. We check now that the formulas for the Hamiltonians are correct.

```
> H01:=simplify(residue( checkL11^2+checkL12*checkL21+checkL12*diff
  (checkL11/checkL12,lambda),lambda=0));
  for j from 0 to 2*d-2 do
  Hinfty[j]:=simplify(-residue(lambda^(-j-1)*(checkL11^2+checkL12*
  checkL21+checkL12*diff(checkL11/checkL12,lambda)),lambda=
  infinity)): od:
  Hinfty[0]:=Hinfty[0];
```

$$H01 := \frac{1}{Qinfty_0}\Big(\big(2\,Qinfty_1\,t_3^2 - 2\,t_2\,t_3\big)\,Qinfty_0^3 + \Big(-Qinfty_1^3\,t_3^2 + 2\,t_2\,Qinfty_1^2\,t_3 + \big(-Pinfty_0^2 - 2\,t_1\,t_3 - t_2^2\big)\,Qinfty_1 - 2\,Pinfty_0\,Pinfty_1 + 2\,t_3\,t_0 + 2\,t_2\,t_1\Big)\,Qinfty_0^2 + t00\,(-1 + t00)\,Qinfty_1\Big)$$

$$Hinfty_0 := \frac{1}{Qinfty_0}\Big(-t_3^2\,Qinfty_0^3 + \big(3\,Qinfty_1^2\,t_3^2 - 4\,Qinfty_1\,t_2\,t_3 - Pinfty_0^2 + 2\,t_1\,t_3 + t_2^2\big)\,Qinfty_0^2 + \Big(-Qinfty_1^4\,t_3^2 + 2\,Qinfty_1^3\,t_2\,t_3 + \big(-2\,t_1\,t_3 - t_2^2\big)\,Qinfty_1^2 + \big(2\,t_3\,t_0 + 2\,t_2\,t_1\big)\,Qinfty_1 + Pinfty_1^2 + Pinfty_0\Big)\,Qinfty_0 + t00^2 - t00\Big) \qquad (15)$$

```
> HamGeneral:=0:
  for k from 0 to 2*d-2 do HamGeneral:=HamGeneral+nu[k+1]*Hinfty[k]
  : od:
  HamGeneral:=HamGeneral+nu[0]*(H01+2*d*Pinfty[2*d-1]):
  for k from 0 to 2*d-2 do HamGeneral:=HamGeneral+nu[0]*(k+1)*
  Qinfty[k+1]*Pinfty[k]: od:
  HamGeneral:=simplify(HamGeneral);
```

$$HamGeneral := \frac{1}{2\,Qinfty_0\,t_3^2}\Bigg(\Big(\big(2\,Qinfty_1\,\alpha_2 - 2\,\alpha_1\big)\,t_3^3 - t_2\,\alpha_2\,t_3^2\Big)\,Qinfty_0^3 + \Bigg(\Big(-Qinfty_1^3\,\alpha_2 + 6\,Qinfty_1^2\,\alpha_1\Big)\,t_3^3 + \bigg(\Big(-Qinfty_1^2\,t_2 - 2\,t_1\,Qinfty_1 + 2\,t_0\Big)\,\alpha_2 - 8\,\alpha_1\,\Big(t_2\,Qinfty_1 - \frac{t_1}{2}\Big)\bigg)\,t_3^2 \qquad (16)$$

$$+\left(\left(\left(-\mathit{Pinfty}_0^2+3\,t_2^2\right)\mathit{Qinfty}_1-2\,\mathit{Pinfty}_0\,\mathit{Pinfty}_1\right)\alpha_2+2\,t_2^2\,\alpha_1-2\,\mathit{Pinfty}_0^2\,\alpha_1\right)t_3+\big($$
$$\mathit{Pinfty}_0^2\,t_2-t_2^3\big)\,\alpha_2\Big)\,\mathit{Qinfty}_0^2+\Big(-2\,\mathit{Qinfty}_1^4\,\alpha_1\,t_3^3+\Big(\mathit{Qinfty}_1^3\,t_2\,\alpha_2+4\,\alpha_1\,\big(\mathit{Qinfty}_1^2\,t_2$$
$$-t_1\,\mathit{Qinfty}_1+t_0\big)\Big)\,\mathit{Qinfty}_1\,t_3^2+\Big(\Big(-2\,\mathit{Qinfty}_1^3\,t_2^2+2\,\mathit{Qinfty}_1^2\,t_1\,t_2+\big(-2\,t_0\,t_2$$
$$+\mathit{Pinfty}_0\big)\,\mathit{Qinfty}_1+2\,\mathit{Pinfty}_1\Big)\,\alpha_2+2\,\alpha_1\,\big(-\mathit{Qinfty}_1^2\,t_2^2+2\,\mathit{Qinfty}_1\,t_1\,t_2+\mathit{Pinfty}_1^2$$
$$+\mathit{Pinfty}_0\big)\Big)\,t_3-t_2\,\alpha_2\,\big(-\mathit{Qinfty}_1^2\,t_2^2+2\,\mathit{Qinfty}_1\,t_1\,t_2+\mathit{Pinfty}_1^2+\mathit{Pinfty}_0\big)\Big)\,\mathit{Qinfty}_0$$
$$+\mathit{t00}\,\Big(\big(\mathit{Qinfty}_1\,\alpha_2+2\,\alpha_1\big)\,t_3-\alpha_2\,t_2\Big)\,(-1+\mathit{t00})\Big)$$

```
> simplify(LQinfty[0]-diff(HamGeneral,Pinfty[0]));
  simplify(LQinfty[1]-diff(HamGeneral,Pinfty[1]));
  simplify(LPinfty[0]+diff(HamGeneral,Qinfty[0]));
  simplify(LPinfty[1]+diff(HamGeneral,Qinfty[1]));
```

$$0$$
$$0$$
$$0$$
$$0 \tag{17}$$

```
> HamGeneralFunction:=unapply(HamGeneral,alpha[1],alpha[2]):
  Hamt1:=simplify(HamGeneralFunction(1,0));
  Hamt2:=simplify(HamGeneralFunction(0,1));
  VectorHam:=Matrix(2*d,1,0):
  VectorHam[1,1]:=1*Hamt1:
  VectorHam[2,1]:=2*Hamt2:
  HVector:=Matrix(2*d,1,0):
  for k from 1 to 2*d-1 do
  HVector[k,1]:=Hinfty[2*d-1-k]: od:
  HVector[2*d,1]:=H01+2*d*Pinfty[2*d-1]:
  for k from 0 to 2*d-2 do HVector[2*d,1]:=HVector[2*d,1]+(k+1)*
  Qinfty[k+1]*Pinfty[k]: od:
  HVector[2*d,1]:=simplify(HVector[2*d,1]):

  simplify(VectorHam-Multiply(M^(-1),HVector));
```

$$\mathit{Hamt1}:=\frac{1}{t_3\,\mathit{Qinfty}_0}\Big(-t_3^2\,\mathit{Qinfty}_0^3+\Big(3\,\mathit{Qinfty}_1^2\,t_3^2+\big(-4\,t_2\,\mathit{Qinfty}_1+2\,t_1\big)\,t_3+t_2^2-\mathit{Pinfty}_0^2\Big)$$
$$\mathit{Qinfty}_0^2+\Big(-\mathit{Qinfty}_1^4\,t_3^2+2\,\mathit{Qinfty}_1\,\big(\mathit{Qinfty}_1^2\,t_2-t_1\,\mathit{Qinfty}_1+t_0\big)\,t_3-\mathit{Qinfty}_1^2\,t_2^2$$
$$+2\,\mathit{Qinfty}_1\,t_1\,t_2+\mathit{Pinfty}_1^2+\mathit{Pinfty}_0\Big)\,\mathit{Qinfty}_0+\mathit{t00}^2-\mathit{t00}\Big)$$

$$\mathit{Hamt2}:=\frac{1}{2\,\mathit{Qinfty}_0\,t_3^2}\Big(\Big(2\,\mathit{Qinfty}_1\,t_3^3-t_2\,t_3^2\Big)\,\mathit{Qinfty}_0^3+\Big(-\mathit{Qinfty}_1^3\,t_3^3+\Big(-\mathit{Qinfty}_1^2\,t_2$$
$$-2\,t_1\,\mathit{Qinfty}_1+2\,t_0\Big)\,t_3^2+\Big(\big(-\mathit{Pinfty}_0^2+3\,t_2^2\big)\,\mathit{Qinfty}_1-2\,\mathit{Pinfty}_0\,\mathit{Pinfty}_1\Big)\,t_3+\mathit{Pinfty}_0^2\,t_2-$$
$$t_2^3\Big)\,\mathit{Qinfty}_0^2+\Big(\mathit{Qinfty}_1^4\,t_2\,t_3^2+\Big(-2\,\mathit{Qinfty}_1^3\,t_2^2+2\,\mathit{Qinfty}_1^2\,t_1\,t_2+\big(-2\,t_0\,t_2+\mathit{Pinfty}_0\big)\,\mathit{Qinfty}_1$$

$$+2\,Pinfty_1\Big)\,t_3+t_2^3\,Qinfty_1^2-2\,t_1\,t_2^2\,Qinfty_1-t_2\left(Pinfty_1^2+Pinfty_0\right)\Big)\,Qinfty_0-t00\,\big(-t_3\,Qinfty_1+t_2\big)\,(-1+t00)\Big)$$

$$\begin{bmatrix}0\\0\end{bmatrix} \quad \textbf{(18)}$$

```
> series(checkL12,lambda);
  series(checkL11,lambda);
  series(checkL21,lambda):
  expand(simplify(residue(checkL21,lambda=0)));
```

$$-Qinfty_0\,t_3-t_3\,Qinfty_1\,\lambda-t_3\,\lambda^2$$

$$t00\,\lambda^{-1}+-Pinfty_1-Pinfty_0\,Qinfty_1+\frac{t00\,Qinfty_1}{Qinfty_0}+\left(\frac{t00}{Qinfty_0}-Pinfty_0\right)\lambda$$

$$t_3\,Qinfty_1^3-2\,Qinfty_0\,t_3\,Qinfty_1-2\,Qinfty_1^2\,t_2+\frac{Pinfty_0^2\,Qinfty_1}{t_3}+2\,Qinfty_0\,t_2+2\,t_1\,Qinfty_1+\frac{t_2^2\,Qinfty_1}{t_3}-\frac{2\,t00\,Pinfty_0\,Qinfty_1}{Qinfty_0\,t_3}+\frac{2\,Pinfty_0\,Pinfty_1}{t_3}-2\,t_0-\frac{2\,t_2\,t_1}{t_3}+\frac{t00^2\,Qinfty_1}{Qinfty_0^2\,t_3}-\frac{2\,t00\,Pinfty_1}{Qinfty_0\,t_3} \quad \textbf{(19)}$$

```
> solve({nu0 = alpha[2]/(2*t[3]) , nu1 = (2*alpha[1]*t[3]-alpha[2]*
  t[2])/(2*t[3]^2)},{alpha[1],alpha[2]});
  alpha[1] := nu0*t[2]+nu1*t[3];
  alpha[2] := 2*nu0*t[3];
  simplify(nu[0]-nu0);
  simplify(nu[1]-nu1);
  checkA12:=simplify(checkA12):
  checkA11:=simplify(checkA11):
  checkA21:=simplify(checkA21):
  series(checkA12,lambda);
  expand(series(checkA11,lambda));
  expand(series(checkA21,lambda));

  simplify(residue(checkA21,lambda=0));
```

$$\{\alpha_1=\nu0\,t_2+\nu1\,t_3,\ \alpha_2=2\,\nu0\,t_3\}$$

$$\alpha_1:=\nu0\,t_2+\nu1\,t_3$$

$$\alpha_2:=2\,\nu0\,t_3$$

$$0$$

$$0$$

$$-\left(\nu0\,Qinfty_1+\nu1\right)t_3\,\lambda-\nu0\,t_3\,\lambda^2$$

$$-\nu 0\, Pinfty_0\, Qinfty_1 + \frac{\nu 0\, t00\, Qinfty_1}{Qinfty_0} - \nu 0\, Pinfty_1 - \nu 1\, Pinfty_0 + \frac{\nu 1\, t00}{Qinfty_0} + \left(-Pinfty_0\, \nu 0 + \frac{\nu 0\, t00}{Qinfty_0}\right) \lambda$$

$$-t_3\, \nu 0\, Qinfty_1^2 + 2\, Qinfty_0\, t_3\, \nu 0 + 2\, \nu 0\, Qinfty_1\, t_2 + 2\, t_3\, \nu 1\, Qinfty_1 + \frac{\nu 0\, Pinfty_0^2}{t_3} - 2\, \nu 0\, t_1 - \frac{\nu 0\, t_2^2}{t_3} - 2\, \nu 1\, t_2 - \frac{2\, \nu 0\, t00\, Pinfty_0}{Qinfty_0\, t_3} + \frac{\nu 0\, t00^2}{Qinfty_0^2\, t_3} + \left(Qinfty_1\, \nu 0\, t_3 - 2\, \nu 0\, t_2 - \nu 1\, t_3\right) \lambda - \nu 0\, t_3\, \lambda^2$$

$$0 \tag{20}$$

```
> expand(series(Hinfty[0],Qinfty[0]));
```

$$\left(t00^2 - t00\right) Qinfty_0^{-1} + -Qinfty_1^4\, t_3^2 + 2\, Qinfty_1^3\, t_2\, t_3 - 2\, Qinfty_1^2\, t_1\, t_3 - Qinfty_1^2\, t_2^2 + 2\, Qinfty_1\, t_0\, t_3 + 2\, Qinfty_1\, t_1\, t_2 + Pinfty_1^2 + Pinfty_0 + \left(3\, Qinfty_1^2\, t_3^2 - 4\, Qinfty_1\, t_2\, t_3 - Pinfty_0^2 + 2\, t_1\, t_3 + t_2^2\right) Qinfty_0 - t_3^2\, Qinfty_0^2 \tag{21}$$

```
> expand(series(H01,Qinfty[0]));
```

$$\left(t00^2\, Qinfty_1 - t00\, Qinfty_1\right) Qinfty_0^{-1} + \left(-Qinfty_1^3\, t_3^2 + 2\, t_2\, Qinfty_1^2\, t_3 - Pinfty_0^2\, Qinfty_1 - 2\, Qinfty_1\, t_1\, t_3 - Qinfty_1\, t_2^2 - 2\, Pinfty_0\, Pinfty_1 + 2\, t_3\, t_0 + 2\, t_2\, t_1\right) Qinfty_0 + \left(2\, Qinfty_1\, t_3^2 - 2\, t_2\, t_3\right) Qinfty_0^2 \tag{22}$$